Two Equations of Lagrange

Bill Page

8/8/2016

1 Two Equations of Lagrange in Terms of Differential Forms

We show that the "two equations of Lagrange" given by Oziewicz and Ramirez are not independent. The second equation is exactly v times the first equation.

This worksheet: https://cloud.sagemath.com/projects/b04b5777-e269-4c8f-a4b8-b21dbe1c93c6/files/Two%

```
%typeset_mode True
```

Consider vectors and forms on a 6-dimensional manifold

```
M = Manifold(5,'M')
coord.<t, x, v, a, s> = M.chart()
f = M.scalar_field(function('f')(*list(coord)))
basis = coord.frame()
[Dt,Dx,Dv,Da,Ds] = [basis[i] for i in range(M.dim())]
cobasis=coord.coframe()
[dt,dx,dv,da,ds] = [cobasis[i] for i in range(M.dim())]
d=xder
def ev(N): return (lambda x: N.contract(x))
```

1.1 For the most general Lagrangian

```
L = M.scalar_field(function('L')(*list(coord))); L.display()
```

$$\begin{array}{llcl} & M & \longrightarrow & \mathbb{R} \\ & (t,x,v,a,s) & \longmapsto & L(t,x,v,a,s) \end{array}$$

f is a dependent variable

```
f = M.scalar_field(var('f'))
```

Kinematics

```
N = Dt + v*Dx + a*Dv + s*Da + f*Ds; N.display()
```

$$\frac{\partial}{\partial t} + v\frac{\partial}{\partial x} + a\frac{\partial}{\partial v} + s\frac{\partial}{\partial a} + f\frac{\partial}{\partial s}$$

```
N(M.scalar_field(t))==1
N(M.scalar_field(x))==v
N(M.scalar_field(v))==a
N(M.scalar_field(a))==s
N(M.scalar_field(s))==f
```

True
True
True
True
True

Define auxillary fields

```
r=Ds(L); r.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & \frac{\partial L}{\partial s} \end{array}$$

```
q=Da(L)-N(r); q.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & -v\frac{\partial^2 L}{\partial x\partial s} - a\frac{\partial^2 L}{\partial v\partial s} - s\frac{\partial^2 L}{\partial a\partial s} - f\frac{\partial^2 L}{\partial s^2} - \frac{\partial^2 L}{\partial t\partial s} + \frac{\partial L}{\partial a} \end{array}$$

```
p=Dv(L)-N(q); p.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & v^2\frac{\partial^3 L}{\partial x^2\partial s} + a^2\frac{\partial^3 L}{\partial v^2\partial s} + 2\,as\frac{\partial^3 L}{\partial v\partial a\partial s} + 2\,af\frac{\partial^3 L}{\partial v\partial s^2} + s^2\frac{\partial^3 L}{\partial a^2\partial s} + 2\,fs\frac{\partial^3 L}{\partial a\partial s^2} + f^2\frac{\partial^3 L}{\partial s^3} + \Big(2\,a \end{array}$$

First Equation

```
(N(p)-Dx(L)).display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & v^3\frac{\partial^4 L}{\partial x^3\partial s} + a^3\frac{\partial^4 L}{\partial v^3\partial s} + 3\,a^2 s\frac{\partial^4 L}{\partial v^2\partial a\partial s} + 3\,a^2 f\frac{\partial^4 L}{\partial v^2\partial s^2} + 3\,as^2\frac{\partial^4 L}{\partial v\partial a^2\partial s} + 6\,afs\frac{\partial^4 L}{\partial v\partial a\partial s^2} + 3 \end{array}$$

Second Equation

```
(Dt(L)-(N(L)-(p*a+q*s+r*f)-(v*N(p)+a*N(q)+s*N(r)))).display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & v^4\frac{\partial^4 L}{\partial x^3\partial s} + \Big(3\,a\frac{\partial^4 L}{\partial x^2\partial v\partial s} + 3\,s\frac{\partial^4 L}{\partial x^2\partial a\partial s} + 3\,f\frac{\partial^4 L}{\partial x^2\partial s^2} + 3\,\frac{\partial^4 L}{\partial t\partial x^2\partial s} - \frac{\partial^3 L}{\partial x^2\partial a}\Big)v^3 + \Big(3\,a^2 \end{array}$$

Multiplying by v

```
v*(N(p)-Dx(L)) == (Dt(L)-(N(L)-(p*a+q*s+r*f)-(v*N(p)+a*N(q)+s*N(r)))\
  )
```

True

1.2 Checking the calculations from the paper

```
L = M.scalar_field(function('L')(*list(coord)))
p = M.scalar_field(function('p')(*list(coord)))
q = M.scalar_field(function('q')(*list(coord)))
r = M.scalar_field(function('r')(*list(coord)))
t=M.scalar_field(t)
x=M.scalar_field(x)
v=M.scalar_field(v)
a=M.scalar_field(a)
s=M.scalar_field(s)
```

Action differential Form

```
alpha = L*dt + p*(dx-v*dt) + q*(dv-a*dt) + r*(da-s*dt)
alpha.display()
```

$$\left(-vp\left(t,x,v,a,s\right)-aq\left(t,x,v,a,s\right)-sr\left(t,x,v,a,s\right)+L\left(t,x,v,a,s\right)\right)\mathrm{d}t+p\left(t,x,v,a,s\right)\mathrm{d}x+q\left(t,x,v,a,s\right)\mathrm{d}v+r\left(t,x,v,a,s\right)\mathrm{d}a$$

```
alpha == L*dt+p*dx+q*dv+r*da-(p*v+q*a+r*s)*dt
```

True

```
d(alpha).display()
```

$$\left(v\frac{\partial p}{\partial x}+a\frac{\partial q}{\partial x}+s\frac{\partial r}{\partial x}-\frac{\partial L}{\partial x}+\frac{\partial p}{\partial t}\right)\mathrm{d}t\wedge\mathrm{d}x+\left(v\frac{\partial p}{\partial v}+a\frac{\partial q}{\partial v}+s\frac{\partial r}{\partial v}+p\left(t,x,v,a,s\right)-\frac{\partial L}{\partial v}+\frac{\partial q}{\partial t}\right)\mathrm{d}t\wedge\mathrm{d}v+\left(v\frac{\partial p}{\partial a}+a\frac{\partial q}{\partial a}+s\frac{\partial r}{\partial a}+q\left(t,x,v,a,s\right)-\frac{\partial L}{\partial a}+\frac{\partial r}{\partial t}\right)\mathrm{d}t\wedge\mathrm{d}a+\left(v\frac{\partial p}{\partial s}+a\frac{\partial q}{\partial s}+s\frac{\partial r}{\partial s}+r\left(t,x,v,a,s\right)-\frac{\partial L}{\partial s}\right.$$
$$\mathrm{d}s+\left(-\frac{\partial p}{\partial v}+\frac{\partial q}{\partial x}\right)\mathrm{d}x\wedge\mathrm{d}v+\left(-\frac{\partial p}{\partial a}+\frac{\partial r}{\partial x}\right)\mathrm{d}x\wedge\mathrm{d}a-\frac{\partial p}{\partial s}\mathrm{d}x\wedge\mathrm{d}s+\left(-\frac{\partial q}{\partial a}+\frac{\partial r}{\partial v}\right)\mathrm{d}v\wedge\mathrm{d}a-\frac{\partial q}{\partial s}\mathrm{d}v\wedge\mathrm{d}s-\frac{\partial r}{\partial s}\mathrm{d}a\wedge\mathrm{d}s$$

```
d(alpha) == d(L).wedge(dt) + d(p).wedge(dx) + d(q).wedge(dv) + d(r).\
    wedge(da) - d(p*v + q*a + r*s).wedge(dt)
```

True

```
ev(alpha)(N)==L
```

True

```
Omega = -(p*dx+q*dv+r*da).wedge(d(t)); Omega.display()
```

$$p\left(t,x,v,a,s\right)\mathrm{d}t\wedge\mathrm{d}x+q\left(t,x,v,a,s\right)\mathrm{d}t\wedge\mathrm{d}v+r\left(t,x,v,a,s\right)\mathrm{d}t\wedge\mathrm{d}a$$

```
alpha == L*dt + ev(N)(Omega)
```

True

Equation of Motion ($E=0$)

```
E = ev(N)(d(alpha))
```

```
E.display()
```

$$\Bigg(-v^2\frac{\partial p}{\partial x} - a^2\frac{\partial q}{\partial v} - as\frac{\partial q}{\partial a} - af\frac{\partial q}{\partial s} - as\frac{\partial r}{\partial v} - s^2\frac{\partial r}{\partial a} - fs\frac{\partial r}{\partial s} - \Bigg(a\frac{\partial p}{\partial v} + s\frac{\partial p}{\partial a} + f\frac{\partial p}{\partial s} + a\frac{\partial q}{\partial x} + s\frac{\partial r}{\partial x} - \frac{\partial L}{\partial x}$$

$$\Bigg(v\frac{\partial p}{\partial x} + a\frac{\partial p}{\partial v} + s\frac{\partial p}{\partial a} + f\frac{\partial p}{\partial s} - \frac{\partial L}{\partial x} + \frac{\partial p}{\partial t}\Bigg) \mathrm{dx} + \Bigg(v\frac{\partial q}{\partial x} + a\frac{\partial q}{\partial v} + s\frac{\partial q}{\partial a} + f\frac{\partial q}{\partial s} + p(t, x, v, a, s) - \frac{\partial L}{\partial v} + \frac{\partial q}{\partial t}\Bigg)$$

$$\Bigg(v\frac{\partial r}{\partial x} + a\frac{\partial r}{\partial v} + s\frac{\partial r}{\partial a} + f\frac{\partial r}{\partial s} + q(t, x, v, a, s) - \frac{\partial L}{\partial a} + \frac{\partial r}{\partial t}\Bigg) \mathrm{da} + \Bigg(r(t, x, v, a, s) - \frac{\partial L}{\partial s}\Bigg) \mathrm{ds}$$

Rewriting it in various ways.

```
E == N(L)*dt - N(t) * d(L) + N(p)*dx - N(x)*d(p) + N(q)*dv - N(v)*d(\
   q) + N(r)*da - N(a)*d(r) - N(p*v+q*a+r*s)*dt + N(t)*d(p*v+q*a+r*s\
   )
```

True

```
E == N(L)*dt - d(L) + N(p)*dx - v*d(p) + N(q)*dv - a*d(q) + N(r)*da \
   - s*d(r) - N(p*v+q*a+r*s)*dt + d(p*v+q*a+r*s)
```

True

```
E == N(L-p*v-q*a-r*s)*dt - d(L - p*v - q*a- r*s) - v*d(p) - a*d(q) -\
    s*d(r) + N(p)*dx + N(q)*dv + N(r)*da
```

True

```
d(p*v) == v*d(p) + p*d(v)
```

True

```
d(q*a) == a*d(q) + q*d(a)
```

True

```
d(r*s) == s*d(r) + r*d(s)
```

True

```
E == N(L - p*v - q*a - r*s)*dt - d(L) + p*dv + q*da + r*ds + N(p)*dx\
    + N(q)*dv + N(r)*da
```

True

```
N(p*v) == p*N(v) + v*N(p)
```

True

```
E == N(L)*dt - (p*a+q*s+r*f)*dt -(v*N(p) + a*N(q) + s*N(r) )*dt - d(\
   L) + p*dv + q*da + r*ds + N(p)*dx + N(q)*dv + N(r)*da
```

True

```
E == ( N(L) - (p*a + q*s + r*f) - (v*N(p) + a*N(q) + s*N(r)))*dt + N\
   (p)*dx + (N(q)+p)*dv + (N(r)+q)*da + r*ds - d(L)
```

True

```
d(L) == Dt(L)*dt + Dx(L)*dx + Dv(L)*dv + Da(L)*da + Ds(L)*ds
```

True

```
r=Ds(L); r.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & \frac{\partial L}{\partial s} \end{array}$$

```
q=Da(L)-N(r); q.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & -v\frac{\partial^2 L}{\partial x\partial s} - a\frac{\partial^2 L}{\partial v\partial s} - s\frac{\partial^2 L}{\partial a\partial s} - f\frac{\partial^2 L}{\partial s^2} - \frac{\partial^2 L}{\partial t\partial s} + \frac{\partial L}{\partial a} \end{array}$$

```
p=Dv(L)-N(q); p.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & v^2\frac{\partial^3 L}{\partial x^2\partial s} + a^2\frac{\partial^3 L}{\partial v^2\partial s} + 2\,as\frac{\partial^3 L}{\partial v\partial a\partial s} + 2\,af\frac{\partial^3 L}{\partial v\partial s^2} + s^2\frac{\partial^3 L}{\partial a^2\partial s} + 2\,fs\frac{\partial^3 L}{\partial a\partial s^2} + f^2\frac{\partial^3 L}{\partial s^3} + \Big(2\,a \end{array}$$

1.3 For example: the Schiff and Poirier Lagrangian

```
hbar = var('hbar',latex_name='\hbar')
m = var('m')
V = M.scalar_field(function('V')(var('x')))
L = 1/2*m*v^2  - V - hbar^2/4/m*(s/v^3-5/2*a^2/v^4); L.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & \frac{4\,m^2v^6 - 8\,mv^4V(x) + 5\,a^2\hbar^2 - 2\,\hbar^2 sv}{8\,mv^4} \end{array}$$

```
r=Ds(L); r.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & -\frac{\hbar^2}{4\,mv^3} \end{array}$$

```
q=Da(L)-N(r); q.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & \frac{a\hbar^2}{2\,mv^4} \end{array}$$

```
p=Dv(L)-N(q); p.display()
```

$$\begin{array}{llll} M & \longrightarrow & \mathbb{R} \\ (t,x,v,a,s,f) & \longmapsto & \frac{4\,m^2v^6 - 2\,a^2\hbar^2 + \hbar^2 sv}{4\,mv^5} \end{array}$$

First Equation

```
(N(p)-Dx(L)).expr().expand()
```

$$am + \frac{5\,a^3\hbar^2}{2\,mv^6} - \frac{2\,a\hbar^2 s}{mv^5} + \frac{f\hbar^2}{4\,mv^4} + D[0]\,(V)\,(x)$$

Second Equation

```
(Dt(L)-(N(L)-(p*a+q*s+r*f)-(v*N(p)+a*N(q)+s*N(r)))).expr().expand()
```

$$amv + vD[0]\,(V)\,(x) + \frac{5\,a^3\hbar^2}{2\,mv^5} - \frac{2\,a\hbar^2 s}{mv^4} + \frac{f\hbar^2}{4\,mv^3}$$